

1 MODULE *MinMax1*

This module and modules *MinMax2* and *MinMax2H* are used as examples in Sections 1 and 2 of the paper “Auxiliary Variables in TLA+”.

This module specifies a tiny system in which a user presents a server with a sequence of integer inputs, and the server responds to each input value i with one of the following outputs: *Hi* if i is the largest number input so far, *Lo* if it’s the smallest number input so far, *Both* if it’s both, and *None* if it’s neither.

The module is part of an example illustrating the use of a history variable. The example includes this module, module *MinMax2*, and module *MinMax2H* which adds a history variable to the specification of *MinMax2* and shows that the resulting specification implements the specification of the current module under a suitable refinement mapping.

19 EXTENDS *Integers*

We define $setMax(S)$ and $setMin(S)$ to be largest and smallest value in a nonempty finite set S of integers.

25 $setMax(S) \triangleq \text{CHOOSE } t \in S : \forall s \in S : t \geq s$

26 $setMin(S) \triangleq \text{CHOOSE } t \in S : \forall s \in S : t \leq s$

The possible values that can be returned by the system are declared to be constants, which we assume are not integers.

32 CONSTANTS *Lo, Hi, Both, None*

33 ASSUME $\{Lo, Hi, Both, None\} \cap Int = \{\}$

The the value of the variable x is the value input by the user or the value output by the system, the variable $turn$ indicating which. The variable y holds the set of all values input thus far. We consider x and $turn$ to be externally visible and y to be internal.

41 VARIABLES $x, turn, y$

42 $vars \triangleq \langle x, turn, y \rangle$

The initial predicate *Init*:

47 $Init \triangleq \wedge x = None$

48 $\wedge turn = \text{“input”}$

49 $\wedge y = \{\}$

The user’s input action:

54 $InputNum \triangleq \wedge turn = \text{“input”}$

55 $\wedge turn' = \text{“output”}$

56 $\wedge x' \in Int$

57 $\wedge y' = y$

The systems response action:

62 $Respond \triangleq \wedge turn = \text{“output”}$

63 $\wedge turn' = \text{“input”}$

64 $\wedge y' = y \cup \{x\}$

65 $\wedge x' = \text{IF } x = setMax(y') \text{ THEN IF } x = setMin(y') \text{ THEN } Both \text{ ELSE } Hi$

66 $\text{ELSE IF } x = setMin(y') \text{ THEN } Lo \text{ ELSE } None$

The next-state action:

71 $Next \triangleq InputNum \vee Respond$

The specification, which is a safety property (it asserts no liveness condition).

77 $Spec \triangleq Init \wedge \Box[Next]_{vars}$

Invariant to check (added by hengxin)

82 $NoneCertificate \triangleq \Box[\wedge x \in Int$
83 $\quad \wedge x < setMax(y \cup \{x\})$
84 $\quad \wedge x > setMin(y \cup \{x\})$
85 $\quad \equiv x' = None]_{vars}$

Below, we check that specification *Spec* implements specification *Spec* of module *MinMax2* under a suitable refinement mapping. The following definitions of *Infinity* and *MinusInfinity* are copied from module *MinMax2*.

93 $Infinity \triangleq \text{CHOOSE } n : n \notin Int$
94 $MinusInfinity \triangleq \text{CHOOSE } n : n \notin (Int \cup \{Infinity\})$

96 $M \triangleq \text{INSTANCE } MinMax2$
97 $\quad \text{WITH } min \leftarrow \text{IF } y = \{\} \text{ THEN } Infinity \quad \text{ELSE } setMin(y),$
98 $\quad \quad \quad max \leftarrow \text{IF } y = \{\} \text{ THEN } MinusInfinity \text{ ELSE } setMax(y)$

The following theorem asserts that *Spec* implements the specification *Spec* of module *MinMax2* under the refinement mapping defined by the *INSTANCE* statement. The theorem can be checked with *TLC* using a model having *M!Spec* as the temporal property to be checked.

106 THEOREM $Spec \Rightarrow M!Spec$

107

\ * Modification History
\ * Last modified Tue Jun 26 17:51:08 CST 2018 by hengxin
\ * Last modified Tue Jun 26 17:29:03 CST 2018 by hengxin
\ * Last modified Fri Oct 21 23:48:10 PDT 2016 by lamport
\ * Created Fri Aug 26 14:28:26 PDT 2016 by lamport